

ASTROVISION ADD-ON is called " ZGRASS-32 COMPUTER KEYBOARD ", and was shown at the Consumer Electronic Show at Las Vegas a month ago. The box, layout of parts, etc., is essentially the same as the one shown by Bally and illustrated on p. 29 of Vol 1. The shown sample was all black, and had an arcade unit resting on its upper surface. What's inside? The following data is taken from a publicity sheet handed out: 32K more RAM, 24K more ROM, the keyboard, and the ZGRASS language. Cost?-\$599. Delivery?-well..I heard 'manufacturing in June' twice, so we'll see. More details...

Language used is Dr. DeFanti's ZGRASS, a graphics-oriented BASIC. A description of the language and its unique commands are contained in Vol 1, pages 11 - 14, and 37. The ZGRASS is contained in the 24K ROM, along with a 'scientific math system'. It gives four colors anywhere on the screen, which is configured at 160x100 pixels. The prime direction for ZGRASS is for the development of graphically exciting programs, whether they be games, video art, pie charts, or whatever. The beginnings were included in the basic arcade unit, with the commands LINE and BOX. New commands will be POINT and CIRCLE. Arrays will be multi-dimensional. There will be interpreted and compiled modes. The extended math package is floating point with trig & log (e and 10) functions, and square root.

PLANNED EXPANSIONS to the keyboard unit include:

32K more RAM

TV PRINTER - hard copy of what is on the TV screen (a scan is made of the screen and a print made of it.)

LIGHT PEN

BIT PAD DIGITIZER - a worksheet is electrically attached to the computer and a wand is touched to the pad. A dot appears on the screen at the same relative point. Drawings can be transcribed to the screen(=computer) and the coordinates of all the points are now in memory

DUAL CASSETTE CONTROL at 1800 baud, motor of cassette is computer-controlled.

DISC STORAGE - a connector is included.

SLIDE COPIER

SOUND SYSTEM - connection to audio amplifier input

More details on the above as they are discovered. One of the reasons for the CES was contact with dealers/distributors, to show the line, and sign up as their plan dictates. Quite a bit has to be regenerated from the old Bally days, as well as expansion beyond those levels to get nation-wide coverage. Regardless, if you are not able to find one in your area (after June,etc.) let me know and we'll handle it by mail order.

NEW GAME VIDEOCADES: Here is the lineup of the first batch - which are now scheduled for March

2011 GALACTIC INVASION / 2012 SPACE FORTRESS / 2014 GRAND PRIX-DEMOLITION DERBY / 4003 MUSIC MAKER 1 / 4004 BIORYTHM

USER GROUP REPORT The CACHE group indicate that they now have a file of 50 programs in their software library. A reference library of printed material (hardware-oriented) is also being set up for the group. It was reported that REX TV has opened a new Bally service center at 18666 S. Dixie Highway, Homewood, IL 779-7800

LISTED PROGRAMS have a number of small letters in them, because my COMPRINT does not have some characters. Therefore, you have to make a little conversion, as follows: a = right arrow > b = multiply * c = divide ÷

AN EXCELLENT TUTORIAL on Machine Language programming - getting some of the basics down - was printed in the November Kilobaud MICROCOMPUTING, p.70. I'd say it was a medium-level description that taught me quite a lot and will be used as a reference in later operations. It explains some of the descriptive codes used (like 8B CA etc). While the article specifically deals with the 6802 processor, the ideas as well as some specifics also fit the Z80.

BLUE RAM OPERATING SYSTEM (1.0) arrived the other day, and it has two very useful features: 1. It allows EDITING of line statements, using a new RPL command (RePLace). You can either fix a line by adding, deleting, or changing something in the line, or you can change the line number from one value to another. People send in programs on tape, and sometimes the transfer to the Bally is not correct (usually a long line) and making the correction is a pain. But now I've used the 1.0 system, and I can easily make the necessary change. Especially since I've loaded the keyboard routine as well. All one needs to do is enter RPL (line number)/(old word)/(new word) RETURN or GO, and it's done. 2. It allows the storage of 3800 Bally BASIC bytes in the Blue Ram. ((NOT machine language, but good ol' BB)) This opens up the world to those of you who aren't too keen on machine language, because all you have to do is follow a few simple rules, and keep writing programs in BB. In essence, there is a divider built in, all program lines below 1999 go into the old Basic storage areas, while material in line numbers above 1999 go into the BR. One can shift back and forth between the two areas. So now there are 5600 bytes available - lets see some programs that take advantage of that space.

ALTERNATIVE ENGINEERING is proceeding with the design and construction of the motherboard memory addition. They report that a small rearrangement has taken place, in that some previous options are now included in the basic package, resulting in about a month's delay. The prime package will include a power supply capable of handling all accessories, p.c. board and connectors, some onboard intelligence to control things, plus some other goodies. The two prime accessories will be the memory package and the keyboard package. Other units are also under consideration. Full details are due in the announcement which I plan to have in the next issue.

EXTENDED BASIC language is moving along smartly - current plans are to provide both tape and ROM versions so you can choose the format that suits you best.

A little preview of the new commands follows:

CIRCLE / POINT / NEW (erases program) / DATA (provides initialization of variables in a shorter manner) / ZERO (sets all variables to zero)

In addition to the usual FC and BC, two more foreground colors are available in FA and FB.

There are commands available for storing and retrieving programs, or just bits of programs.

Two windows can be generated on the screen, and their size limits controlled, for both graphics and characters.

How about a 3x5 character set available at CF=SMALL ?

And it will automatically convert to a hex number base with the ! character. SCROLL will roll the screen up or down a specified number of lines.

> in front of a variable will give its memory location.

That ought to whet your appetite - we'll have more information as it becomes available, including delivery and cost, in subsequent issues.

The following tutorial was written by Bob Wiseman, and details some of his programming 'tricks', explaining why.

PART I. HOW TO KEEP IT SMALL.

WHEN YOU PROGRAM THE ARCADE, YOU HAVE A GRAND TOTAL OF 1.8K OF USER MEMORY AVAILABLE FOR PROGRAM AND STRING DATA. THIS IS SELDOM ENOUGH. WHENEVER I PROGRAM THE ARCADE, I WRITE THE PROGRAM ON PAPER FIRST. THIS ALLOWS ME TO CALCULATE MEMORY REQUIREMENTS EARLY SO THAT I CAN HEAD 'EM OFF AT THE PASS. I STRUCTURE THE PROGRAM INTO MODULES WHENEVER POSSIBLE. I WRITE EACH MODULE ON A SEPARATE SHEET SO THAT I CAN REWRITE A SINGLE MODULE WITHOUT AFFECTING ANYTHING ELSE. HAVING A PENCIL COPY OF A PROGRAM IS AN INVALUABLE AID WHEN SEARCHING FOR THAT ELUSIVE BUG.

SO, I WRITE THE PROGRAM ON PAPER, AND THEN I ADDUP THE MEMORY NEEDED. I HAVE 1800 BYTES TO START WITH.

COUNT THE FOLLOWING:

- 3 BYTES PER LINE FOR STORAGE FOR THE LINE NUMBER AND CARRAGE RETURN.
- 1 BYTE PER KEYWORD. THE WORDS PRINT, LIST, GOTO, LINE, FOR, TO, NEXT, STEP, GOSUB, RETURN, BOX, INPUT IF, AND RND ALL TAKE ONE BYTE.
- 1 BYTE PER PUNCTUATION. THESE SPECIAL CHARACTERS TAKE ONE BYTE: COMMA, PERIOD, QUOTE, EXCLAMATION, PARANTHESIS, EQUALS, POUND SIGN, ETC..
- 1 BYTE PER ALPHABETIC LETTER. EACH LETTER, WHETHER USED AS A VARIABLE OR USED IN QUOTES, COUNTS 1 BYTE. ITEMS LIKE CX, BC, AND NT HAVE TWO LETTERS AND SO THEY COUNT TWO BYTES. SPACE COUNTS ONE BYTE.
- 1 BYTE FOR EACH NUMERIC SYMBOL NOT BEGINNING A LINE.

HERE ARE SOME EXAMPLES:

10 GOTO 10	(COUNTS 6 BYTES)
20 PRINT "ABC	(COUNTS 8 BYTES)
300 IF A=RND (3)GOTO 400	(COUNTS 14 BYTES)

IN ADDITION TO THE MEMORY USED IN YOUR PROGRAM, YOU MAY NEED TO PROVIDE SPACE FOR STRINGS OF DATA. I AM REFERRING TO THE USE OF THE AT-SIGN TABLE TO STORE DATA. EACH TAKES TWO BYTES. SO IF YOU USE @ (0), @ (1),... , @ (9), THEN YOUR PROGRAM MUST LEAVE TWENTY BYTES FREE.

THE FOLLOWING SHORT PROGRAM:

```
10 FOR A=0TO9:@(A)=0:NEXTA
```

WILL TAKE 19 BYTES TO STORE, AND AN ADDITIONAL TWENTY BYTES WHEN IT IS RUN FOR STORAGE OF THE DATA.

NOW THAT WE KNOW WHERE THE MEMORY GOES, WE SHOULD BE ABLE TO USE IT MORE EFFICIENTLY. THESE GENERAL RULES MAY HELP.

- 1) COMBINING SHORT LINES TOGETHER HELPS REDUCE THE THREE BYTES PER LINE OVERHEAD.
- 2) ELIMINATE UNNECESSARY SPACES. THE ONLY TIME THAT A SPACE IS REQUIRED BY BALLY BASIC IS TO SEPARATE TWO VARIABLES. FOR EXAMPLE:
 10 IF A=CGOTO 10 (NO SPACES REQUIRED)
 20 IF B=D E=F (SPACE REQUIRED TO SEPARATE D AND E)
- 3) USE VARIABLES TO REPLACE OFTEN USED, LONG NUMERIC VALUES. THIS CAN BE ESPECIALLY VALUABLE FOR LINE NUMBERS. EACH TIME YOU USE THE NUMERIC VALUE 3210 (FOR EXAMPLE), YOUR PROGRAM IS FOUR BYTES LARGER. IF YOU SAY "T=3210" THAT COSTS SIX BYTES. IF YOU NEED TO REFER TO 3210 MORE THAN TWICE (AND YOU HAVE AN AVAILABLE VARIABLE OF COURSE) YOU CAN SAVE MEMORY.

EXAMPLE:

10 IF A=3210B=B+1	5 T=3210
20 IF B=3210C=C+1	10 IFA=TB=B+1
30 IF C=3210GOTO10	20 IFB=TC=C+1
	30 IF C=TGOTO10

THE PROGRAM ON THE LEFT TAKES 15+15+13= 43 BYTES. THE ONE ON THE RIGHT TAKES 9+12+12+10= 43 BYTES. THE ONE ON THE RIGHT RUNS FASTER. IF WE HAD PUT "T=3210" ON LINE 10, THEN THE RIGHTHAND PROGRAM WOULD BE SMALLER ALSO.

- 4) OMIT TRAILING QUOTATION MARKS. LINE TEN AND TWENTY WILL PRINT THE SAME THING.

EXAMPLE:

```
10 PRINT "ABCDEF"
20 PRINT "ABCDEF"
```

- 5) DO NOT BE AFRAID TO RE-WRITE. FIRST, YOU WRITE WHAT WORKS. NEXT, YOU WRITE WHAT WORKS BETTER. FINALLY, YOU WRITE IT SMALLER.

EXAMPLE: SUPPOSE WE HAVE REACHED A POINT IN THE PROGRAM WHERE WE HAVE A ZERO OR A ONE IN VARIABLE "A". IF ONE, WE WANT TO CHANGE "A" TO A ZERO. IF ZERO, WE WANT TO CHANGE "A" TO A ONE. SO WE FIRST WRITE THE OBVIOUS:

```
100 IF A=0A=1:GOTO120
110 A=0
120 ...CONTINUE
```

THIS WORKS AND IT TAKES 15+6= 21 BYTES. FURTHER HEADSCRATCHING AND THE MEMORY BULB LIGHTS UP:

```
100 A=1-A
```

THIS TAKES A MERE 8 BYTES AND DOES THE SAME THING.

I HAVE OFTEN DISCOVERED THAT FINDING A SIMPLE ALGEBRAIC FORMULA WILL SAVE MANY BYTES. THE DIFFICULT PART IS FINDING THE FORMULA.

- 6) USE SUBROUTINES WHENEVER POSSIBLE. ANYTHING THAT YOUR PROGRAM DOES MORE THAN ONCE SHOULD BE IN A SUBROUTINE. WHEN YOU BREAK UP A PROGRAM INTO SUBROUTINES IT BECOMES EASIER TO DEBUG BECAUSE THE SUBROUTINES CAN BE TESTED INDEPENDENT OF THE REST OF THE PROGRAM. DRAWING A SHORT FLOWCHART (HORRORS!) BEFORE BEGINNING CODING IN BASIC WILL HELP ORGANIZE YOUR THOUGHTS ABOUT WHAT SHOULD BE A SUBROUTINE. SUBROUTINES HELP AVOID REDUNDANT CODE, AND THIS KEEPS IT SMALL.

PART II. MAKING IT RUN FASTER.

THE BEST TIME TO MAKE A PROGRAM RUN FASTER IS BEFORE YOU HAVE WRITTEN IT. AFTERWARDS, THE CHANGES MAY BE TOO DIFFICULT TO MAKE AND DEBUG (SORT OF LIKE ADDING AN EXTRA BATHROOM ON THE SECOND FLOOR AFTER THE HOUSE HAS BEEN BUILT). THE GENERAL RULE I HAVE FOLLOWED IS:

(OTHER THINGS BEING EQUAL) THE FEWER BYTES BALLY BASIC HAS TO INTERPRET THE FASTER THE PROGRAM RUNS.

THIS EQUATES TO PART I. IN GENERAL, THE SHORTER A PROGRAM, THE FASTER IT RUNS. I HAVE RUN A FEW SHORT TIMINGS AND THEY SEEM TO FOLLOW THIS RULE. REMEMBER OUR EXAMPLE WHERE WE REPLACED THE VALUE 3210 WITH THE VARIABLE "T". THE EXAMPLE ON THE RIGHT RUNS FASTER BECAUSE BASIC HAS FEWER BYTES TO INTERPRET.

HERE IS AN EXPERIMENT FOR YOU TO PERFORM. WRITE THE BASE PROGRAM AS FOLLOWS:

```
10 FOR A=0 TO 3000
90 NEXT A
```

RUN THIS AND TIME ITS EXECUTION WITH A WATCH. RECORD THIS TIME. NOW TRY INSERTING DIFFERENT LINES INTO THE CENTER AND SEE HOW THIS AFFECTS THE TOTAL RUN TIME. I FOUND THIS TO BE QUITE INTERESTING. IF NO GRAPHICS ARE INVOLVED, I FOUND THAT THE EXTRA EXECUTION TIME WAS ALMOST DIRECTLY PROPORTIONAL TO THE EXTRA NUMBER OF BYTES INTERPRETED. USE EACH OF THE FOLLOWING AND SEE HOW THEIR TIMES COMPARE.

```
20 B=A          (6 BYTES)
20 B=12345+1    (12 BYTES)
```

I FIND THAT BOX AND LINE REQUIRE MORE TIME PER INTERPRETED BYTE THAN NON-GRAPHIC COMMANDS.

IF YOU CANNOT MAKE IT RUN FAST, YOU MIGHT AS WELL MAKE IT ENTERTAINING. TRY TYING SOME OF YOUR VARIABLES TO THE SOUND SYSTEM INPUTS. A MUSICAL DIVERSION WILL LET THE OPERATOR KNOW THAT THE PROGRAM IS STILL WORKING, AND HASN'T DROPPED DEAD OR FOUND A LOOP.

ARCADIAN


```

4 .GOLF
5 .BY BOB HENSEL
100 FOR Z=1TO N;@(Z)=0;NEXT Z;B=0;FOR H=1TO 9;GOSUB 3600
102 FOR P=1TO N;U=0;J=0
104 IF P=1G=13+RND (50);I=15+RND (15);T=13+RND (50);R=13+RND (50);S=13+RND (50)
;B=B+Ic5
105 CLEAR ;GOSUB 1000;Y=-16;GOSUB 1100;T=T+18;GOSUB 1100;T=T-18;Y=0;GOSUB 1200
106 Y=10;GOSUB 1300;E=40;F=-41;BOX E,F,1,1,1
142 LINE 3,-43,4;LINE 3,43,1;LINE 79,43,1;LINE 79,-43,1;LINE 3,-43,1
144 PRINT " HOLE #",#1,H;PRINT " PAR ",#1,Ic5," ",#1,Ib20,"YD"
146 W=RND (4);M=(RND (6)-1)b5;PRINT " WIND:",;TV=93+W;PRINT #3,M,"MPH",
147 CY=8;CX=-71;PRINT "PLAYER SCORE",
148 CX=-56;CY=0;PRINT #1,P,#7,U
150 A=A+JX(1);IF A>16A=1
151 IF A<1A=16
152 X=2;Y=2;GOSUB 800+A
153 LINE -40,-20,4;BOX -40,-20,60,30,2;BOX -40,-20,2,2,1;LINE Xb7-40,Yb7-20,1
155 C=C+JY(1);IF C>9C=1
156 IF C<1C=9
170 CX=-71;CY=-40;PRINT "CLUB:",#1,C;IF C=1PRINT " WOOD",;GOTO 180
172 PRINT " IRON",
180 IF TR(1)=0GOTO 150
200 U=U+1;NT=10;MU=C;NT=0
202 IF J=1GOTO 3100
205 FOR Z=1TO 2b(10-C)
210 IF Z#11GOTO 219
212 D=0;IF M>5D=1
214 IF W=1Y=Y+D
216 IF W=2X=X-D
217 IF W=3Y=Y-D
218 IF W=4X=X+D
219 E=E+X;F=F+Y;BOX E,F,1,1,3
220 IF E>3IF E<79IF F<43IF F>-43J=0;GOTO 240
230 GOTO 4000
240 IF E>G-9IF E<G+8IF F>I-8IF F<I+8J=1
250 IF J=0IF PX(E,F)=0GOTO 4000
290 NEXT Z
291 IF J=1GOTO 3000
295 GOTO 148
801 Y=0;RETURN
802 Y=1;RETURN
803 RETURN
804 X=1;RETURN
805 X=0;RETURN
806 X=-1;RETURN
807 X=-2;RETURN
808 X=-2;Y=1;RETURN
809 X=-2;Y=0;RETURN
810 X=-2;Y=-1;RETURN
811 X=-2;Y=-2;RETURN
812 X=-1;Y=-2;RETURN
813 X=0;Y=-2;RETURN
814 X=1;Y=-2;RETURN
815 Y=-2;RETURN
816 Y=-1;RETURN
1000 BOX G,I,4,12,1;BOX G,I,10,10,1;BOX G,I,14,8,1;BOX G,I,16,4,1;RETURN
1100 BOX T,Y,2,10,1;BOX T,Y+2,4,2,1;BOX T,Y,6,2,1;BOX T,Y-2,8,2,1;RETURN

```

Bob Hensel
8428 Ingleside Ave., S.
Cottage Grove, MN 55016

Golf is a game of skill for 1 to 4 players. The computer generates 9 different holes each game, randomly placing each green and hazards such as trees, water, or sand traps. Each player selects the direction the ball will travel by moving JX(1) until the rotating line points in the proper direction. The distance is dependent on the club selected using JY(1). Remember the flight of the ball is affected by the direction and velocity of the wind.




COLOR ORGAN/LIGHT SHOW: Interface Bally to any Stereo Amp. with audio power of 1 watt or up. Will not effect Stereo Quality. TV displays Color & Light in step with your Stereo.

Complete interface unit with Cassette Demo. and Tutorial ..\$29.95 (Add your own Art/Graphics).

Many Games are now available- ALL HAVE FULL MEMORY usage, Lots of Graphics, Action and Color. Any two (2) Games on Cassette, with Documentation \$ 10.00

STEREO
COLOR ORGAN
ON YOUR TV
INTERFACES WITH
BALLY ARCADE
ASTROVISION

ALL COLORS
BRILLIANT DISPLAY



GAMES : THE MUMMYS TREASURE
AYATOLLAH DART BOARD
SINK THE U-BOAT
RESCUE AIR DROP
AIR RAID
TARGET (TRIO OF GAMES)
KILL THE VULCAN
SEARCH & DESTROY

BOMBARDIER
CRAZY BALL
STAR BASE 2000
THE BLACK LAGOON
MISSION IMPOSSIBLE
SPACE QUEST 2001
FOX AND THE HARE
GALACTIC WAR 2002
ATOM SMASHER

WRITE FOR FREE CATALOG:

BLANK COMPUTER CASSETTES @ \$1.00 ea.

GOLF (continued)

```
1200 BOX R,Y,2,18,1;BOX R,Y,6,14,1;BOX R,Y,10,10,1;BOX R,Y,14,6,1;BOX R,Y,18,2,1
;RETURN
1300 BOX S,Y,12,5,1;RETURN
3000 BOX 41,0,78,87,2;BOX 41,0,16,48,1;BOX 41,0,40,40,1;BOX 41,0,56,32,1;BOX 41,
0,72,16,1
3010 K=20+RND (36);L=RND (16)-8;BOX K,L,3,3,2;E=41+(E-G)*4;F=(F-I)*4;BOX E,F,1,1
,3;BOX 41,0,79,84,3
3015 IF ABS(E-K)<2IF ABS(F-L)<2GOTO 3500
3017 GOTO 148
3100 FOR Z=1TO 26(10-C)
3110 E=E+X;F=F+Y;BOX E,F,1,1,3
3120 IF ABS(E-K)<2IF ABS(F-L)<2GOTO 3500
3130 IF E>3IF E<79IF F<43IF F>-43GOTO 3140
3135 GOTO 4000
3140 NEXT Z;GOTO 148
3500 CY=0;CX=30;PRINT "SUNK!";@(P)=@(P)+U
3510 GOSUB 3610;NEXT P;NEXT H
3600 CLEAR ;PRINT " PAR=",#1,B;FOR Z=1TO N;PRINT " PLAYER",#2,Z,"=",#1,@(Z),#4,@
(Z)-B;NEXT Z
3610 FOR O=1TO 1000;NEXT O;RETURN
4000 CY=-20;CX=-59;PRINT "PENALTY";NT=40;MU=60;MU=60;MU=60;NT=0;U=U+1;GOSUB 3610
;GOTO 148
```



```

1 .
2 .
3 .
4 .
5 .MUSICAL STAFF
6 .BY BOB WISEMAN
10 GOSUB 950
20 N=N+1
30 GOSUB 300; IF W GOTO 50
40 GOSUB 200; GOTO 20
50 GOSUB 400; X=X-9
60 C=JX(1)+2bJY(1)
70 IF C=0 GOTO 60
80 IF C=-2 GOTO 10
90 FOR M=0 TO N-1; P=@(M)c50; T=RM
100 GOSUB 700; NEXT M
110 IF C=2 IF TR(1)=0 GOTO 120
115 GOTO 30
120 FOR M=0 TO 100; NEXT M; GOTO 85
200 @(N)=50bP+T; RETURN
300 X=X+9; IF X>70 X=-70; Y=15-Y; GOSUB 500
310 GOSUB 800; GOSUB 700
320 IF TR(1)+TR(2)W=TR(2); RETURN
330 V=JX(1); W=JY(1)
340 IF V=0 IF W=0 U=0; GOTO 320
350 IF U=1 GOTO 320
360 U=1; P=P+W; IF (P>19)+(P<0)P=P-W
365 IF V=0 GOTO 385
370 IF V<0 T=Tc2
380 IF V>0 T=T+T
385 IF T>32 T=32
387 IF T<2 T=2
390 GOSUB 400; GOTO 310
400 BOX X+1, Y+7, 10, 12, 2
405 BOX X+1, Y-27, 10, 14, 2
410 FOR V=Y-17 TO Y-2 STEP 5
415 BOX X+2, V, 10, 4, 2; NEXT V
420 RETURN
500 BOX 0, Y-10, 160, 44, 2
505 FOR V=Y-20 TO Y STEP 5
510 BOX 0, V, 160, 1, 1; NEXT V
515 RETURN
700 @(16)=49; @(17)=@(B-P); @(22)=127
710 FOR A=0 TO 20bT; NEXT A
715 @(22)=0
790 RETURN

```

```

800 I=0; IF P=Pc2b2I=1
810 V=Y-(20-P)b5c2+16
815 BOX X, V, 5, 4, 1
818 IF (P<5)+(P>15)GOSUB 900
825 IF T>12 BOX X, V, 3, 1, 2; IF I=0 BOX X, V-1, 3, 1, 2
830 IF T=32 RETURN
835 H=5-I
845 IF P>11 H=-H-2+I
847 K=2bABS(H)-1
850 BOX X+2, H+V, 1, K, 1
855 IF T>6 RETURN
856 IF H<0 K=2-K

```

MUSICAL STAFF writes notes in the key of C, using controller #1. JY command changes the note, while JX command changes the length. TR enters the note. Controller #2 plays the tune once, repeatedly, or starts over.

```

858 IF T>3 IF P>11 GOTO 870
860 BOX X+3, K+V, 3, 1, 1
865 IF T>3 RETURN
870 BOX X+3, K+V-2, 3, 1, 1; RETURN
900 IF P>15 BOX X, Y+5, 8, 1, 1
905 IF P>17 BOX X, Y+10, 8, 1, 1
910 IF P<5 BOX X, Y-25, 8, 1, 1
915 IF P<3 BOX X, Y-30, 8, 1, 1
920 RETURN

```

Bob Wiseman
118 St. Andrews Dr.
Cincinnati OH 45245

```

950 : RETURN ; NT=0; CLEAR
951 M=SZc2; N=M; A=1000; B=90; GOSUB A; B=80; GOSUB A
952 B=71; GOSUB A; B=67; GOSUB A; B=60; GOSUB A; B=53; GOSUB A; B=50; GOSUB A
953 B=44; GOSUB A; B=39; GOSUB A; B=35; GOSUB A; B=33; GOSUB A; B=29; GOSUB A; B=26; GOSUB A
954 B=24; GOSUB A; B=22; GOSUB A; B=19; GOSUB A; B=17; GOSUB A; B=16; GOSUB A; B=14; GOSUB A
960 P=4; T=8; N=-1
961 B=M
965 X=70; Y=-10; RETURN
1000 N=N-1; @(N)=B; RETURN

```


A selection of short programs from the programming efforts of
Dieter Heinerman, 505 4th Ave.S.W. #1511 Calgary, Alberta, Canada T2P 0J8

```

1 .
2 .
3 . 3D CORNERS
4 . BY DIETER HEINERMAN
9 NT=0
10 CLEAR
11 FC=7
16 FOR A=1TO 44STEP 1
17 IF TR(1)=1GOTO 50
18 IF TR(2)=1CLEAR
19 BC=7bB
20 BOX A, -A, A, A, 3
25 BOX -A, A, A, A, 3
30 BOX A, A, A, A, 3
34 BOX -A, -A, A, A, 3
35 &(22)=255; &(18)=A
36 NEXT A
37 B=RND (255)
39 FC=B
40 GOTO 16
50 FOR A=44TO 1STEP -2
70 GOTO 20

```

```

2 .
3 . ELECTRONIC BLANKED
4 . BY DIETER HEINERMAN
5 NT=0
9 CLEAR
10 A=RND (6)
11 FOR E=1TO 5
12 .TRY NT=A
13 &(22)=255
15 IF E=5GOTO 90
20 FOR B=-70TO 70STEP A
25 &(17)=B; &(18)=BbA
30 C=RND (6)
35 IF TR(1)=1GOTO 9
40 FOR D=-40TO 40STEP C
41 MU=BC
42 BC=B+200
43 FC=BC-53bC
45 &(16)=DbC
46 &(20)=A+CbD
50 BOX B, D, A, A+C, 3
60 NEXT D
70 NEXT B
75 NEXT E
80 GOTO 10
90 PRINT "    WANT MORE? PRESS 1"
100 IF &(23)=8GOTO 9
110 GOTO 100

```

```

1 .
2 .
3 .
4 . DENOMINATOR
5 . BY DIETER HEINERMAN
6 CLEAR
7 BC=56
8 NT=1
9 PRINT ;PRINT " INPUT DENOMINATOR
10 INPUT " ", N
20 FOR T=1TO N-1
30 X=0
40 PRINT
50 A=T
60 PRINT #2, A, #1, "c", #1, N, #1, "=.",
70 IF (Ab10)cN<1GOTO 140
80 A=Ab10
90 PRINT #1, (AcN),
100 X=X+1
110 IF X>=N-1GOTO 190
120 A=A-(AcN)bN
130 GOTO 70
140 A=Ab10
150 PRINT #1, 0,
160 X=X+1
170 IF X>=N-1GOTO 190
180 GOTO 70
190 NEXT T
200 IF &(23)=1GOTO 1
210 GOTO 200

```

```

1 .BOXES
2 .
3 .
4 .
5 NT=10
6 CLEAR
7 BC=127
10 &(22)=255
20 FOR C=1TO 100STEP 1
30 FC=BbA
40 A=RND (140)-70
50 B=RND (80)-40
55 MU=A
56 &(18)=B
57 &(20)=C
60 BOX A, B, 5, 5, 3
61 BOX -A, B, 5, 5, 3
62 BOX A, -B, 5, 5, 3
65 BOX -A, -B, 5, 5, 3
70 IF C>99GOTO 90
80 NEXT C
90 FOR E=1TO 10STEP 1
95 MU=E
100 IF E=10CLEAR
110 IF TR(1)=1GOTO 10
120 NEXT E
130 GOTO 90

```


ADS:WANTED: Used Bally interface in good working condition - cheap.

R. Ruppert 2722 Pyramid Ave., Pittsburgh PA 15227

FOR SALE: Software by W&W Software Sales, 6594 Swartout Rd. Algonac MI 48001. Nine tapes with five programs each, plus a new one for the Blue Ram (This one is available on tape only, \$5., and called 'Bally Nuclear Power Plant') See ad on p. 10 for more details...

FOR SALE: Cartidges because UPS lost my Arcade: BASIC, INTERFACE at \$35 each; BASEBALL/FOOTBALL/SEA WOLF/CLOWNS/MATH BINGO/280ZZZAP/STAR BATTLE/RED BARON/BLACKJACK at \$15.each. Check or money order plus \$1. to Randy Rienth, Rt 1 Box 73H, Gloucester, VA 23061

FOR SALE: Software by Rob Rosenhouse, 44 Forestbrook Dr., North Plainfield, NJ 07060. Super Software now offers 6 sets of programs including a new one based on Missile Command by Atari. For catalogue including 2 RND(ART) programs, send a large SSAE

USER GROUPS: If you are in the following areas, contact the persons listed:
Metropolitan Washington, DC-Jerry Heere, 2802 Avon Ave Silver Spring 19608
Westchester Co.areaNY-Dan Simpson Box 229 Somers, NY 10589
914-248-7058

FOR SALE: Bally factory games and accessories - special discount to ARCADIAN subscribers. For free price list, write to SFP, 1064 N. Alta Ave. Dinuba, CA 93618

2 AND 3-PART HARMONY

**MUSIC
SOFTWARE!**

Three 60-minute programming
cassettes now available!

1. *Bach's 15 Two-Part Inventions!*
 2. *Christmas Music; 27 Carols!*
 3. *Scott Joplin Ragtime Piano!*
- \$9⁵⁰ Each**



GEORGE MOSES CO.

P.O. Box 686 • Brighton, MI 48116

**HOT FLASH! LATEST
ASTRO VISION PRICE LIST
SHOWS AUGUST DATE FOR
ADD-ON.**

≡ 50 ≡

ARCADIAN

Robert Fabris, yellow ribboned
The SOURCE TCD 959
3626 Morrie Dr.
San Jose, CA 95127

FIRST CLASS